
Abstract
JSL, JMP’s powerful scripting language, can be used to extend JMP’s features in a variety of 
ways. But when your JSL script has an international audience, what is the best way to localize 
its user interface? This paper describes the tools and methodology used by the JMP 
development team to localize the JSL scripts that are built-in to JMP. You can combine these 
same tools with your own methodology to give your scripts an international appeal.

Introduction
Advanced JMP users construct workflows and applications in JSL. Often, these JSL scripts gain 
a wide audience, if they are found to be useful. In some cases, the audience for a JSL script can 
include users who run JMP in a locale that is different from what the script was originally written 
for. If the script contains its own user interface elements, this can be disconcerting for the 
international user, who experiences JMP’s interface in their own language, but must run a script 
that presents its interface in a foreign tongue.

As a simple example, here is a JSL script that prompts the user for some input parameters, then 
generates a data table of random character strings. It exists in a file named 
RandomTextGenerator.jsl. Here is the first half of the script.

r = New Window( "Random Text Generator", <<Modal, 

 <<On Validate( 
  ncol = Num( ncol box << Get Text ); 
  nrow = Num( nrow box << Get Text ); 
  wlen = Num( wlen box << Get Text ); 
  alpha = alpha box << Get Text; 
  1 
 ), 

 V List Box( 
  Lineup Box( N Col( 2 ), Spacing( 10, 15 ), 
   Text Box( "Number of columns:" ), ncol box = Text Edit Box( "5" ), 
   Text Box( "Number of rows:" ),  nrow box = Text Edit Box( "100" ), 
   Text Box( "Word length:" ),   wlen box = Text Edit Box( "10" ), 
   Text Box( "Alphabet:" ),    alpha box = Text Edit Box( 
    "BCDFGHJKLMNPQRSTVWXZ" ) 
  ), 
  H List Box( Spacer Box(), Button Box( "Cancel" ), Button Box( "OK" )) 
 ) 
); 
If( r["Button"] == -1, Return()); 



When run, it collects the parameters in a modal window which looks like this.

When the OK button is clicked, the second half of the script runs.

dt = New Table( "Random Text" ); 
For( c = 1, c <= ncol, c++, 
 cname = Substitute( "Column ^I", "^I", Char( c )); 
 dt << New Column( cname, Character ); 
); 

dt << Add Rows( nrow ); 

For( r = 1, r <= nrow, r++, 
 For( c = 1, c <= ncol, c++, 
  col = Column( dt, c ); 

  word = ""; 
  For( w = 1, w <= wlen, w++, 
   word ||= Substr( alpha, Random Uniform( 1, Length( alpha ) + 1 ), 1 ); 
  ); 

  col[ r ] = word; 
 ) 
); 

It generates a data table which looks something like this.

There are several opportunities where elements of this script could better accommodate an 
international user, if they were localized. Certainly, all the text in the modal window could be 
translated, including the window’s title and its “OK” and “Cancel” buttons. Even in the generated 
data table, its title and column names might be candidates for translation.

The Wrong Way
There are a few options for how we can achieve the goal of localizing this script. One possibility 
is that we simply send the script as-is to our translators, ask them to determine which strings 
require translation, and have them send us back a new copy of the script with the translated 
strings inserted.

This works great for the script author, at least initially. It has very little start-up cost. But let’s list 
the ways that this system breaks down over time.

• It is not scalable. This system can grow on two axes: the number of distinct scripts in your 
system × the number of languages to which you must translate. In our example, we are only 
translating a single script. For one script, this system might work; but what if we have 5 
scripts, or 10? Now this solution is not scaling well.

• It is not maintainable. We have duplicated our script, once for each language, and translated 
the strings. But now we need to fix a bug, or add a feature, or even simply rename a variable. 
We must replicate our changes in each copy of the script, and make sure that all the copies 
remain synchronized. We have increased the chance of introducing a maintenance error.

• It asks too much of translators. The person writing the script is typically not the same person 
who must translate it to other languages. Asking the translator to edit JSL correctly might be 
outside of their skill set. The format of a JSL script might not be compatible with the 
translation tools the translator uses. Even asking the translator to choose which strings 
require translation invites confusion, as one translator might make a choice that’s different 
from another translator.

• It asks too much of the users. When a user wants to run the script, they must now choose 
which of the 7 or so variants of the script they would like to run. If they choose wrong, they 
are presented with a user interface that they may not be able to interpret. The script should 
automatically present the correct strings for the user’s locale.

The JMP Way
JMP provides a better (though until now undocumented) way. It is used internally by parts of 
JMP that are implemented with JSL scripts, like the tutorials on the Help menu. To use it, you 
simply add this line at the top of your script.

Include( "=Locale.jsl" ); 

This includes one of JMP’s built-in scripts. (The “=” character at the beginning of the name tells 
JMP to load it from the built-in scripts folder.) The built-in Locale.jsl script does several things. 
First, it determines the file from which it was included, and looks for a corresponding file in the 
same folder that contains translated strings.

In our example, it would find RandomTextGenerator.jsl and look for a file in the same folder 
named RandomTextGenerator.xlf. An .xlf file contains XLIFF , the XML Localization Interchange 1

File Format, a standard format for representing localized text that is commonly used by 
translators and their tools. Each translatable string in our script can be represented in the XLIFF 
file as a translation unit (“trans-unit”), that contains not only the original string but its translation 
in each language.

Locale.jsl reads the .xlf file into a JSL associative array, and provides access via the function 
Locale:get(). Give this function a string ID, and it will return that string’s appropriate translation 
for the current locale. So now we must “mark up” our script to indicate which strings we want to 
translate, using Locale:get(). Here’s the script again, this time with the translatable strings 
indicated.

Include( "=Locale.jsl" ); 

r = New Window( Locale:get( "Random Text Generator" ), <<Modal, 

 <<On Validate( 
  ncol = Num( ncol box << Get Text ); 
  nrow = Num( nrow box << Get Text ); 
  wlen = Num( wlen box << Get Text ); 
  alpha = alpha box << Get Text; 
  1 
 ), 

 V List Box( 
  Lineup Box( N Col( 2 ), Spacing( 10, 15 ), 
   Text Box( Locale:get( "Number of columns:" )), 
    ncol box = Text Edit Box( "5" ), 
   Text Box( Locale:get( "Number of rows:" )), 
    nrow box = Text Edit Box( "100" ), 
   Text Box( Locale:get( "Word length:" )), 
    wlen box = Text Edit Box( "10" ), 
   Text Box( Locale:get( "Alphabet:" )), 
    alpha box = Text Edit Box( "BCDFGHJKLMNPQRSTVWXZ" ) 
  ), 
  H List Box( Spacer Box(), 
   Button Box( Locale:get( "Cancel" )), Button Box( Locale:get( "OK" ))) 
 ) 
); 
If( r["Button"] == -1, Return()); 

dt = New Table( Locale:get( "Random Text" )); 
For( c = 1, c <= ncol, c++, 
 cname = Substitute( Locale:get( "Column ^I" ), "^I", Char( c )); 
 dt << New Column( cname, Character ); 
); 

// etc., no changes in the remainder of the script 

 See https://en.wikipedia.org/wiki/XLIFF.1

There are several things to point out here.

• We added an Include for Locale.jsl at the top.

• Most text that is visible to the user is wrapped within the Locale:get() function. For example, 
we did this for each Text Box label that is beside a Text Edit Box where the user can type.

• We did not wrap the Text Edit Box strings, which become the initial values in those boxes. 
These strings will not be translated.

• There are other strings that are not translated, such as the string “Button” which we use to 
determine what button was used to dismiss the dialog. These strings are internal to the 
operation of our script and are not visible to the user.

• When constructing the data table’s column names, we use the Substitute() function to place 
calculated values into the column names. This allows us to provide a single “pattern string” for 
translation, which is preferable to simply building up the string using concatenation. The 
translator can also choose to rearrange the substitution token(s) as needed for proper word 
order in their target language.

JMP has provided us with the run-time framework for localization. If we now run this script, the 
result is identical to our previous version. That’s because there is no RandomTextGenerator.xlf 
file yet, and Locale:get() simply returns its argument if it cannot find a string with the requested 
ID.

A Word About String IDs
I’ve mentioned the concept of a string ID a few times, but I haven’t yet defined what this is. 
Here’s an example of why it is important. Imagine you have a script that creates two windows, 
both of which have the title “Train”. When run, you get this.

Obviously, these two window titles should be translated differently! But if you are using the direct 
English to identify a trans-unit, you must use the same trans-unit for both window titles. Instead, 
XLIFF allows you to create an ID for each trans-unit that is separate from the trans-unit’s actual 
source text. The ID is any arbitrary string, and is unique for each trans-unit. If you assign string 
IDs to the trans-units in your XLIFF file, then you can specify them to Locale:get() when you 
need a string.

Next Steps
Next we need to build an XLIFF file that contains trans-units for each of the Locale:get() strings 
in our script. We should also generate unique string IDs for each string, and modify the script to 
use those string IDs instead of the direct English strings. We could construct this file by hand, 
but that would be extremely tedious. It would be better to have an automated tool that builds the 
XLIFF file. JMP does not provide such a tool.

But I have provided just such a tool that can be downloaded with this paper. It is, of course, 
written in JSL, and I call it ExtractStrings.jsl. The method I use is effective, but it does have its 
shortcomings. However, it makes for a good starting point for building your own localization 
methodology.

The script, as provided, is already configured to process RandomTextGenerator.jsl. But it can be 
easily changed to process a different script, or extended to process an entire folder of scripts. 
When run, it creates a RandomTextGenerator.xlf, which is ready for localization to German. It 
generates unique string IDs for each string, and it rewrites RandomTextGenerator.jsl to use 
those string IDs.

Here’s a snippet of the generated XLIFF file.

<?xml version="1.0" encoding="UTF-8"?> 

<!-- ======================================================== --> 
<!-- Strings for 3. Extracted/RandomTextGenerator.jsl --> 
<!-- Copyright © 2017 MyCompany, Inc. All Rights Reserved. --> 
<!-- ======================================================== --> 

<xliff version="1.2" 
 xmlns="urn:oasis:names:tc:xliff:document:1.2" 
 xmlns:jmp-xstr="http://www.jmp.com/xmlns/jmp-xstr.xsd"> 

 <file source-language="en" original="RandomTextGenerator.jsl" 
  datatype="plaintext" xml:space="preserve"> 
  <header> 
   <jmp-xstr:version>1</jmp-xstr:version> 
   <jmp-xstr:lang>de;en</jmp-xstr:lang> 
  </header> 
  <body> 

  <group resname="All"> 
   <trans-unit id="S_Random_Text_Generator_Win" restype="window"> 
    <source>Random Text Generator</source> 
    <alt-trans alttranstype="reference" xml:lang="de"> 
     <target state="x-new">Random Text Generator</target></alt-trans> 
   </trans-unit> 

We see that the first trans-unit in the XLIFF file is the string used as the modal window title. 
ExtractStrings.jsl assigned it a string ID of “S_Random_Text_Generator_Win”, using JMP’s 
internal naming convention for such things. But you could modify the string ID generator to suit 
your specific needs.

If we look at the first few lines of our script, we see that it was also rewritten to use this string ID.

Include( "=Locale.jsl" ); 

r = New Window( Locale:get( “S_Random_Text_Generator_Win” ), <<Modal, 

Again, we can run this modified script and not see anything different. Even though all the 
infrastructure is now in place, we haven’t actually performed any translations yet! But, we see 
that the trans-units contain an alt-trans block for German. This is where we would put our 
translations.

So we send this XLIFF file to our German translator, who loads it into their translation tool which 
supports XLIFF (or they can even edit it by hand). They return to us a file where the German alt-
trans blocks contain their work. Now, when we run our script in the German locale, we see this 
user interface .2

 Please forgive my naive German translations.2

Clicking OK gives us a data table similar to before, but now the table name and the column 
names are properly translated.

Note that the translation modifies the placement of the numbering in the column names. This is 
enabled by our use of the Substitute() function to build the strings.

Conclusion
JMP’s built-in Locale.jsl tool provides solid run-time support for script localization. For some, the 
ExtractStrings.jsl script provided with this paper provides a sufficient localization methodology. 
But there are a few ways it could be better.

• The script does its work by using the parse only mode of the JSL Include() function. This 
parses the target script into an expression tree which it then recursively walks through looking 
for Locale:get() functions. After editing the tree, it streams it back out to overwrite the target 
script. But doing so loses any comments and formatting contained in the original script.

• Substitution strings like the one used for column names in the example could be better 
marked up in the XLIFF file. This would allow an XLIFF-savvy translation tool to treat the 
substitution string as a token, and prevent the translator from accidentally translating it.

• For large projects being translated to multiple languages by multiple independent translation 
teams, a more detailed methodology might be needed. For example, the JMP development 
team uses additional tools to split the XLIFF file for each script into separate files for each 

translation team, then merge the results back together as they are returned to us. The tools 
also automatically check out scripts and XLIFF files from our source management system, 
then check back in the translated results once they are recombined.

Contact information

